/******************************************************************************
 * Copyright (c) 2025, NVIDIA CORPORATION.  All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *     * Redistributions of source code must retain the above copyright
 *       notice, this list of conditions and the following disclaimer.
 *     * Redistributions in binary form must reproduce the above copyright
 *       notice, this list of conditions and the following disclaimer in the
 *       documentation and/or other materials provided with the distribution.
 *     * Neither the name of the NVIDIA CORPORATION nor the
 *       names of its contributors may be used to endorse or promote products
 *       derived from this software without specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND
 * ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED
 * WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE
 * DISCLAIMED. IN NO EVENT SHALL NVIDIA CORPORATION BE LIABLE FOR ANY
 * DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES
 * (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES;
 * LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND
 * ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
 * (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
 * SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
 *
 ******************************************************************************/

#pragma once

#include <cub/detail/ptx-json/string.h>
#include <cub/detail/ptx-json/value.h>

#include <cuda/std/__type_traits/integral_constant.h>

namespace ptx_json
{
template <auto K, typename V>
struct keyed_value
{
  __forceinline__ __device__ static void emit()
  {
    value<K>::emit();
    asm volatile(":" ::: "memory");
    V::emit();
  }
};

template <typename T>
struct is_keyed_value : cuda::std::false_type
{};

template <auto K, typename V>
struct is_keyed_value<keyed_value<K, V>> : cuda::std::true_type
{};

template <typename T>
concept a_keyed_value = is_keyed_value<T>::value;

template <auto... KV>
struct object;

template <>
struct object<>
{
  __forceinline__ __device__ static void emit()
  {
    asm volatile("{}" ::: "memory");
  }
};

template <a_keyed_value auto First, a_keyed_value auto... KVs>
struct object<First, KVs...>
{
  __forceinline__ __device__ static void emit()
  {
    asm volatile("{" ::: "memory");
    First.emit();
    ((comma(), KVs.emit()), ...);
    asm volatile("}" ::: "memory");
  }
};

template <typename T>
struct is_object : cuda::std::false_type
{};

template <auto... KV>
struct is_object<object<KV...>> : cuda::std::true_type
{};

template <string V>
struct key
{
  template <typename U>
  __forceinline__ __device__ constexpr keyed_value<V, U> operator=(U)
  {
    return {};
  }
};
} // namespace ptx_json
